Tutorial 1

Week of September 19, 2005

1 CFGs

Give CFG grammars (using BNF notation) for the following languages or say it cannot be done

1. All strings over the alphabet {0,1,2,3} sorted in decreasing order.

Answer:

<S> ::= <threes> <twos> <ones> <zeros> | epsilon
<threes> ::= 3 <threes> | epsilon
<twos> ::= 2 <twos> | epsilon
<ones> ::= 1 <ones> | epsilon
<zeros> ::= 0 <zeros> | epsilon

2. All strings over the alphabet {a,b}, in which every b is both immediately preceded by and followed by at least one a. Answer:

<S> ::= a <BA> <S> | epsilon
<BA> ::= b a <BA> | epsilon

3. a^n b^m a^n b^m n, m >= 0

cannot be expressed by CFG

2 english descriptions

Give English descriptions of the languages described by each of the following grammars.

a) <S> ::= <S> a <S> b <S> | <S> b <S> a <S> | epsilon

   Answer:
          All strings with an equal number of a’s and b’s.

   Example:
     <S> ==> <S> a <S> b <S>
         ==> <S> b <S> a <S> a <S> b <S>
         ==> <S> b <S> aa <S> b <S>
         ==> <S> b <S> aa <S> b <S> a <S> b <S>
         ==> baabab

b) <S> ::= a <S> a | b <S> b | c <S> c | epsilon

   Answer: All even length strings over {a,b,c}, where the
           second half of the string is the reverse of the first half.

   Example:
    <S> ==> a <S> a
        ==> a b <S> b a
        ==> a b c <S> c b a
        ==> a b c b <S> b c b a
        ==> abcb bcba

c) Adjust the grammar in (b) to make it generate all palindromes.

   What do we need? We need to also generate odd length strings.

   Answer: <S> ::= a <S> a | b <S> b | c <S> c | a | b | c | epsilon

3 ambiguity

a) <S> ::= <S> and <S> | <S> or <S> | a | b | c

    Is this grammar ambiguous?
    Yes. Two disticnt parse trees for string ‘‘a and b or c’’.

    Develop a grammar that generates the same strings,
    but is not ambiguous.

    <S> ::= <OR>
    <OR> ::= <AND> | <OR> or <AND>
    <AND> ::= <VAR> | <AND> and <VAR>
    <VAR> ::= a | b | c

b) <S> ::= a <S> a | b <S> b | epsilon

  Not ambiguous.